

;\*\*\*\*\* VAL-Routine

```
VAL      CALL  &CEFB  ;Ausdruck holen
         CALL  &DD37  ;Test auf
         DB    &29    ;Klammer zu
         PUSH  HL      ;Programmzeiger retten
         CALL  GETVAL ;Value berechnen
         POP   HL
         RET
```

```
GETVAL   CALL  &FBDA  ;Stringparameter holen
         JP    Z,&FF0A;Akkuinhalt übernehmen
         EX    DE,HL   ;HL=Stringadresse
         PUSH  HL      ;merken
         LD    D,0
         LD    E,A     ;Länge nach DE
         ADD   HL,DE
         LD    E,(HL)  ;Ende sichern
         LD    (HL),D  ;0 einsetzen
         EX    (SP),HL;HL=Stringadresse
         PUSH  DE
         CALL  &DEBB  ;-> Interpretercode
         CALL  &CEFB  ;Ausdruck berechnen
         POP   DE
         POP   HL
         LD    (HL),E  ;Restore Ende
         JP    &FF2D  ;Test auf numerisches
                        Ergebnis
```

;\*\*\*\*\* GETKEY\$-Routine

```
key$    call &ce67      ;8 Bit-Wert holen
        ld    b,a      ;nach B
        call &dd37      ;Test auf
        db    &29      ;")"
        push  hl       ;Programmzeiger retten
        call &bb81      ;TXT CURSOR ON
read     call &bb09      ;KM READ CHAR
        jr    c,read    ;Tastaturpuffer leeren
        ld    c,0      ;Zeichenzähler auf 0
        ld    hl,&aca4   ;Start Eingabepuffer
        push  hl       ;merken
wait     call &bb06      ;KM WAIT CHAR
        cp    13       ;ENTER gedrückt ?
        jr    z,ready   ;ja, fertig
        ld    e,a      ;Zeichen nach E
        ld    a,b      ;maximale Länge
        cp    c        ;erreicht ?
        jr    z,wait    ;ja, nicht übernehmen
        ld    a,e      ;Zeichen auf Bildschirm
        call &bb5a      ;TXT OUTPUT
        ld    (hl),a    ;und in den Eingabepuffer
        inc   hl        ;Zeiger und
        inc   c         ;Zähler erhöhen
        jr    wait      ;auf Zeichen warten

ready    call &bb84      ;TXT CURSOR OFF
        ld    a,c      ;Stringlänge nach A
        call &fc19      ;Platz reservieren
        pop   hl        ;Start Eingabepuffer
        ld    b,0
        or    a         ;Leerstring ?
        jr    z,leerst
        ldir                ;String übertragen
leerst   call &fbba      ;String übernehmen
        pop   hl        ;Programmzeiger zurück
        ret
```